# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-claripy
_pkgname=${pkgname#python-}
pkgver=9.2.182
pkgrel=1
pkgdesc='An abstraction layer for constraint solvers.'
arch=('any')
url='https://pypi.org/project/claripy/#files'
license=('custom:unknown')
depends=('python' 'z3' 'python-future' 'python-cachetools' 'python-decorator'
         'python-pysmt')
makedepends=('python-build' 'python-pip')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/$_pkgname-$pkgver.tar.gz")
sha512sums=('855e6e34768500cf2703c9f31d800fd0c0f839ca2a0ce15f4b5a8814b0a55013213a9d64ae0590f15f6d8b6f1a355967a595441ed8c4f614dabc41901c84210f')

build() {
  cd "$_pkgname-$pkgver"

  python -m build --wheel --outdir="$startdir/dist"
}

package() {
  cd "$_pkgname-$pkgver"

  pip install \
    --verbose \
    --disable-pip-version-check \
    --no-warn-script-location \
    --ignore-installed \
    --no-compile \
    --no-deps \
    --root="$pkgdir" \
    --prefix=/usr \
    --no-index \
    --find-links="file://$startdir/dist" \
    $_pkgname
}

